Partial Derivatives and Taylor Series and Plotting for $f(x, y)$

```
> f := x^2-y^2-x*y+2*x;
```

$$f := x^2 - x\,y - y^2 + 2\,x$$

The partial derivative with respect to x and y are given by

```
> diff(f,x);
```

$$2\,x - y + 2$$

```
> diff(f,y);
```

$$-x - 2\,y$$

The second partial derivatives

```
> diff(f,x,x);
```

$$2$$

```
> diff(f,y,y);
```

$$-2$$

```
> diff(f,x,y);
```

$$-1$$

```
> diff(f,y,x);
```

$$-1$$

To graph a function of two variables use the plot3d command.
The syntax is **plot3d**($f(x, y)$, $x = a..b, y = c..d,$ *optional parameters*).
Here is a graph of $f(x, y) = x^2 + y^2$ a paraboloid.

```
> plot3d( x^2+y^2, x=-2..2, y=-2..2 );
```



The following examples show some of the options. To see all the options see

```
> ?plot3d, options
> plot3d( exp(-x^2-y^2), x=-3..3, y=-3..3, style=hidden, axes=
  frame, labels=[x,y,z] );
```




```
> f ;
```

$$x^2 - x\,y - y^2 + 2\,x$$

```
> plot3d( f, x=-3..3, y=-3..3, style=patchcontour, contours=21,
  title="bell curve" );
```



To solve f(x,y) = 2 we can do this visually by graphing $f(x, y)$ and the plane 2 and seeing where they intersect

```
> plot3d( [f,2], x=-3..3, y=-3..3, style=patchcontour, color=
  [default, red] );
```



If we want to study the partial derivatives at a point $x = a, y = b$ it is helpful to identify the point. I did this by drawing a spike, that is, a vertical line through the point (a, b). First, it is helpful to use functional notation so I can write $f(1, 1)$.

```
> (a,b) := (1,1);
```

$$a, b := 1, 1$$

```
> f := (x,y) -> x^2-x*y-y^2+2*x;
```

$$f := (x, y) \rightarrow x^2 - x\,y - y^2 + 2\,x$$

```
> f(a,b);
```

$$1$$

```
> f(x,y);
```

$$x^2 - x\,y - y^2 + 2\,x$$

```
> with(plots);
```

[*animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, densityplot, display, dualaxisplot, fieldplot, fieldplot3d, gradplot, gradplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, intersectplot, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra_supported, polyhedraplot, rootlocus, semilogplot, setcolors, setoptions, setoptions3d, spacecurve, sparsematrixplot, surfdata, textplot, textplot3d, tubeplot*]

```
> spacecurve( [a,b,z], z=-5..15, color=blue, thickness=5 );
```



Now, to draw the splike together with the surface we assign the plots to variables and display them together

```
> P := plot3d( f(x,y), x=-3..3, y=-3..3, style=patchcontour,
  contours=21 );
```

$$P := PLOT3D(...)$$

```
> S := spacecurve( [a,b,z], z=-5..15, color=blue, thickness=5 );
```

$$S := PLOT3D(...)$$

```
> display([P,S]);
```



The tangent lines through the point (a, b) are given by

$Tx = f(a, b) + f_x(a, b) \cdot (x - a)$ and $Ty = f(a, b) + f_y(a, b) \cdot (y - a)$

and the tangent plane is given by

$Txy = f(a, b) + f_x(a, b) \cdot (x - a) + f_y(a, b) \cdot (y - a)$

```
> fx := diff( f(x,y), x );
```

$$fx := 2\,x - y + 2$$

```
> fy := diff( f(x,y), y );
```

$$fy := -x - 2\,y$$

```
> fxab := eval(fx,{x=a,y=b});
```

$$fxab := 3$$

```
> fyab := eval(fy,{x=a,y=b});
```

$$fyab := -3$$

```
> Tx := f(a,b) + 3*(x-a);
```

$$Tx := -2 + 3\,x$$

```
> Ty := f(a,b) - 3*(y-b);
```

$$Ty := 4 - 3\,y$$

```
> Txy := Tx + Ty - f(a,b);
```

$$Txy := 1 + 3\,x - 3\,y$$

```
> Px := spacecurve( [x,b,Tx], x=-3..3, color=red, thickness=5 );
```

$Px := PLOT3D(...)$

```
> Py := spacecurve( [a,y,Ty], y=-3..3, color=blue, thickness=5 );
```

$Py := PLOT3D(...)$

```
> display( [P,Px,Py] );
```




```
> Pxy := plot3d( Txy, x=-1..3, y=-1..3, color=yellow ,lightmodel=
  light1);
```

$Pxy := PLOT3D(...)$

```
> display( [P,Px,Py,Pxy] );
```



15
10
5
0
-5
-10
-3
-2
-1
0
1
2
y
-3
-2
-1
0
1
2
3
x